

CORRECTION ALGO-GAME

1--nombre mystère--

VARIABLES

```
nombreMystere EST_DU_TYPE NOMBRE
nombreUtilisateur EST_DU_TYPE NOMBRE
tour EST_DU_TYPE NOMBRE
```

DEBUT_ALGORITHME

```
//génération d'un nombre aléatoire
nombreMystere PREND_LA_VALEUR ALGOBOX_ALEA_ENT(1,100)
//on boucle tant que le nombre mystère est différent de
celui entré
TANT_QUE (nombreUtilisateur != nombreMystere) FAIRE
    DEBUT_TANT_QUE
        //On compte les tours en gardant dans une variable
        tour PREND_LA_VALEUR tour + 1
        LIRE nombreUtilisateur
        SI (nombreMystere < nombreUtilisateur) ALORS
            DEBUT_SI
                AFFICHER "Le nombre mystère est plus petit"
            FIN_SI
        SI (nombreMystere > nombreUtilisateur) ALORS
            DEBUT_SI
                AFFICHER "Le nombre mystère est plus grand"
            FIN_SI
        SI (nombreMystere == nombreUtilisateur) ALORS
            DEBUT_SI
                AFFICHER* "BRAVO !!"
                AFFICHER "Vous avez trouvé le nombre mystère en "
+ tour + " essais"
            FIN_SI
    FIN_TANT_QUE
```

FIN_ALGORITHME

2—pierre feuille ciseaux--

VARIABLES

```
choix EST_DU_TYPE NOMBRE
ia EST_DU_TYPE NOMBRE
fin EST_DU_TYPE NOMBRE
choixIa EST_DU_TYPE CHAINE
choixJoueur EST_DU_TYPE CHAINE
```

DEBUT_ALGORITHME

```
AFFICHER* "choisissez un nombre :"
AFFICHER* "1-Pierre"
AFFICHER* "2-Feuille"
AFFICHER* "3-Ciseaux"

TANT_QUE (fin==0) FAIRE
    DEBUT_TANT_QUE
        //Génération d'un choix aléatoire
        ia PREND_LA_VALEUR random()*(3-1)+1
        ia PREND_LA_VALEUR round(ia)

        SI (ia == 1) ALORS
            DEBUT_SI
                choixIa PREND_LA_VALEUR "Pierre"
            FIN_SI
        SI (ia == 2) ALORS
            DEBUT_SI
                choixIa PREND_LA_VALEUR "Feuille"
            FIN_SI
        SI (ia == 3) ALORS
            DEBUT_SI
                choixIa PREND_LA_VALEUR "Ciseaux"
            FIN_SI
        LIRE choix
        SI (choix == 1) ALORS
            DEBUT_SI
                choixJoueur PREND_LA_VALEUR "Pierre"
            FIN_SI
        SI (choix == 2) ALORS
            DEBUT_SI
                choixJoueur PREND_LA_VALEUR "Feuille"
            FIN_SI
        SI (choix == 3) ALORS
            DEBUT_SI
                choixJoueur PREND_LA_VALEUR "Ciseaux"
            FIN_SI
        SI (choix==ia) ALORS
            DEBUT_SI
                AFFICHER* "Ia a choisi " + choixIa
```

```

        AFFICHER* "Vous avez choisi " + choixJoueur
        AFFICHER* "Match null, on recommence"
    FIN_SI
    SI (choix==1&ia==2 OU choix==2&ia==3 OU
choix==3&ia==1) ALORS
        DEBUT_SI
        AFFICHER* "Ia a choisi " + choixIa
        AFFICHER* "Vous avez choisi " + choixJoueur
        AFFICHER* "Vous avez perdu"
        fin PREND_LA_VALEUR 1
        FIN_SI
    SI (choix==1&ia==3 OU choix==2&ia==1 OU
choix==3&ia==2) ALORS
        DEBUT_SI
        AFFICHER* "Ia a choisi " + choixIa
        AFFICHER* "Vous avez choisi " + choixJoueur
        AFFICHER* "Vous avez Gagné!"
        fin PREND_LA_VALEUR 1
        FIN_SI
    FIN_TANT_QUE
FIN_ALGORITHME

```

3—Jeux des bâtonnets--

VARIABLES

```

nbBats EST_DU_TYPE NOMBRE
batsRestant EST_DU_TYPE NOMBRE
batsActuel EST_DU_TYPE NOMBRE
joueur EST_DU_TYPE NOMBRE
tour EST_DU_TYPE NOMBRE

```

DEBUT_ALGORITHME

```

tour PREND_LA_VALEUR 1
batsActuel PREND_LA_VALEUR 16
//Affichage des bats
POUR nbBats ALLANT_DE 1 A batsActuel
    DEBUT_POUR
    AFFICHER " | "
    FIN_POUR
//Definition de la fin de partie
TANT_QUE (batsActuel>0) FAIRE
    DEBUT_TANT_QUE
    //Définition des tours des joueurs
    SI (tour%2==0) ALORS
        DEBUT_SI
        AFFICHER* ""

```

```

        AFFICHER* ""
        AFFICHER* "Tour du Joueur 2"
        FIN_SI
    SINON
        DEBUT_SINON
        AFFICHER* ""
        AFFICHER* ""
        AFFICHER* "Tour du Joueur 1"
        FIN_SINON
    AFFICHER* ""
    AFFICHER* "Retirez 1, 2 ou 3 baton"
    LIRE joueur
    //Définition des erreurs de choix
    SI (joueur>3 OU joueur<1) ALORS
        DEBUT_SI
            AFFICHER* "vous devez choisir 1,2 ou 3"
        FIN_SI
    SINON
        DEBUT_SINON
            //Affichage des batons après retrait
            batonsActuel PREND_LA_VALEUR batonsActuel -
joueur
                POUR batonsRestant ALLANT_DE 1 A batonsActuel
                    DEBUT_POUR
                        AFFICHER "|"
                    FIN_POUR
                FIN_SINON
            tour PREND_LA_VALEUR tour+1
        FIN_TANT_QUE
        //Vérification du perdant
        SI (tour%2!=0) ALORS
            DEBUT_SI
                AFFICHER* "Le joueur 2 a perdu"
            FIN_SI
        SINON
            DEBUT_SINON
                AFFICHER* "Le joueur 1 a perdu"
            FIN_SINON
    FIN_ALGORITHME

```